
RANDOM_ROM Manual 
HP-41 Module  

 
 

Introduction and Credits.  

Welcome to the Random ROM (“RANROM” for short), yet another HP-41 little adventure around the 

whimsical realm of aleatory confines, where we should have some fun learning about sequences of 

random numbers and how to determine the goodness of those mystifying little buggers, for the lack of 

a better term ;-). 

As the name implies this module gathers a collection of routines and utilities about Random Numbers 

on the HP-41 platform. This includes a few MCODE pseudo-random number generators (p-RNG) from 

diverse sources like the (never released!) Toulouse Math ROM; Journal contributions (eventually 

coalescing into the the SandMath module); as well as a ported version of the Voyager implementation. 

As a second category, other MCODE and RPN-based p-RNGs are also included, both from the 

PPC/Datafile archives and Jean-Marc Baillard’s collection.  

Besides the expected set of utilities and small routines on the random topic, an attempt has been made 

to have a self-contained group of routines and programs to make the content as complete as possible - 

within the space constraints imposed by the ROM format. To that effect, a couple of UPL programs are 

included to evaluate the different p-RNGs in terms of their randomicity, normality, etc. Due credit is 

given to the original programmers in the respective sections of this manual. 

Special thanks are due to Valentín Albillo who suggested numerous enhancements and additional 

subjects to include, such as Gaussian-distributed p-RNGs and others. He also provided critical feedback 

on several sections and steered the development towards a wider range of subjects that have no doubt 

shaped up the module for the better. Make sure you don’t miss the adaptation of his brilliant 

“Mandelbrot Set Area Estimation” from his HP Collection.  

On the programming side, thanks to Mark Power and Håkan Thörgren for their classic contributions in 

Datafile and PPC. Credit is also due to “Mike (Stgt)” for porting the HP’s M-code featured in the 

Voyager series (HP-11C / HP-15C). And lastly, thanks to the programmers of the Toulouse Math ROM 

and the CCD Module for their seminal work on the subject. 

Caveat Lector: Not being an expert on the field I have used this project as a learning vehicle myself, 

hoping that the discovery path and final result can also be interesting to other people as well. Whether 

it has worked or not I can say that at least I tried, but of course you are free to do your very own ;-) 

 

Dependencies. 

This ROM is designed for the HP-41CX O/S, obviously housed in Q-RAM-capable hardware devices like 

Clonix/NoVRAM, MLDL_2k and others. As a general rule no additional software dependency exists, so it 

will also run on any CX-equivalent system such as the SY-41CL (with a TIME module) and the DM-41X. 

The exception to this rule is the Mandelbrot Set Area, which needs the 41Z Module - and in turn the 

Library#4 as well. 

Without further ado, here is a list of the functions in the Main FAT table. 

 

XROM# Function Description Author 

06.00 -RANDOM 1C Section header n/a 

06.01 SEEDT Sets initial Seed / Manual or Time-based Håkan Thörgren 

09.02 RNDM RNG Sequence from SEEDT Håkan Thörgren 

06.03 1RAN Time-based initial Seed  JM Baillard 

06.04 RAN00 RNG from seed in R00 Mark Power 

06.05 RAN20 RNG from seed in R20 Mark Power 

06.06 RANR _ _ RNG from seed in prompt register Power-Martin 

06.07 RCLSD Recalls Voyager Seed HP Co./ Mike (Stgt) 

06.08 RN RN from SEED HP Co./ Mike (Stgt) 

06.09 SEED Sets initial Seed HP Co./ Mike (Stgt) 

06.10 RAND RNG from SEED Toulouse Math ROM 

06.11 RANDXY RNG formatted by XY Toulouse Math ROM 

06.12 STORAND Stores new initial seed Toulouse Math ROM 

06.13 -MONTECARLO Section header n/a 

06.14 AINT Append integer X Frits Ferwerda 

06.15 IROUND Integer Round Ángel Martin 

06.16 “MCE MC-based calculation for e Valentín Albillo 

06.17 “MCPI MC-base calculation for p Albillo-Martin 

06.18 “MCLN2 MC-base calculation for Ln2 Albert Chan 

06.19 “MCLN2+ Driver for MCLN2 ÁM 

06.20 “MBA Mandelbrot Set Area Estimation Albillo-Martin 

06.21 “MCITG+ Driver for MCITG Ángel Martin 

06.22 “MCITG MC Integration functions one variable ÁM 

06.23 “FX Example function got MCITG – 1 variable ÁM 

06.24 “MCITG3 MC Multiple Integration (up to thee vars) Ángel Martin 

06.25 “FXY Example function for MCITG2 – 2 variables ÁM 

06.26 “FXYZ Example function for MCITG – 3 variables ÁM 

06.27 “MCITGN MC Multiple Integration – n variables Greg McClure 

06.28 -RANDOMNESS Section header n/a 

06.29 BXMR Gaussian RNG using Box Muller  Ángel Martin 

06.30 ERF Error function Baillard-Martin 

06.31 MREV Mantissa Digit Reversal Ángel Martin 

06.32 RANG Gaussian RNG using “12R-6” Ángel Martin 

06.33 DGT Mantissa Digit Sum Ángel Martin 

06.34 “EVAL Evaluation of p-RNGs L. H. Gilbert 

06.35 “RNG1 p-RNG Sequence 1 JM Baillard 

06.36 “RGN2 p-RNG Sequence 2 JM Baillard 

06.37 “RNG3 p-RNG sequence 3 JM Baillard 

06.38 “RNG4 p-RNG Sequence 4 JM Baillard 

06.39 “STRAT Stratified Random Sampling ravi – MoHP forum 

06.40 “BENCH Benchmarking Gaussian RNGs Ángel Martin 

06.41 “TRANG Testing Gaussian RNGs Ángel Martin 

06.42 “12R-6 “12x minus 6” Method Ángel Martin 

06.43 “BX-MR Box Muller Method Ángel Martin 

XROM# Function Description Author 

06.44 “INDEX Deviation Index from Normal distribution Ángel Martin 

06.45 “TSTRNG Testing Integer RNG’s Charles T. Tart 

06.46 “CHI Chi-Square Test AUG 1982, 10 pgs. 

06.47 “UC1 Ulam’s Conjecture – V1 Robert G. Wilson 

06.48 “UC2 Ulam’s Conjecture – V2 Ward Edwards 

06.49 “UC3 Ulam’s Conjecture – V3 Gerhard Kruse 

 

 

 

 

 

 

 

 

 

The contents in a nutshell: 

SandMath Random Numbers 

The first set of RNG functions is taken from the SandMath, which used versions of Håkan Thörngren’s 

p-RNGs published in the PPC Calculator Journal. The functions are: 

Function Description Input Output 

SEEDT Sets initial seed  Value in X Loaded in buffer 

RNDM Randon Mumber Current RN in buffer Next RN in the sequence 

 

SEEDT  takes the fractional part of the value in X as initial seed for the RN’s sequence. If the value in 

X is zero then the function will use the actual Time & Date information to generate the seed, assuming 

of course that the Time Module is present (not a problem when using a HP-41CX). 

The value is saved in the first register (right above the header) of Buffer #9, thus it’s compatible with 

the OSX, SandMath and CCD versions of the same functionality. 

On the other hand, RNDM generates the next random number in the sequence, which obeys to the 

following LCG rule (albeit using 0 for the final term, ‘mod 0”): 

r(k+1) = FRC [ r(k) * 9,821 + 0.211327 ] 

well-known to PPC members, since it was also used in the RN routine – albeit using a regular data 

register instead of a buffer for the actual storage of the sequence terms. 

Variations on a theme.- 

The formula above is good because it already provides the RNs in canonical form. i.e. their values are 

between 0 and 1 (this one excluded). It however shows a flagrant weakness in that the last three 

decimal digits are always zero. You can check this using the PPC ROM, the CCD Module or previous 

versions of the SandMath itself (*), as the three use the same LCG expression (Linear Congruent 

Generator). 

In the RANROM I have used a different approach, simply overwriting the last three digits with the first 

three of the mantissa in reversed order. Now, you may argue (and probably will be right) that this isn’t 

an orthodox way to tackle the problem but intuitively it’s got to be better than the three-zero case, thus 

the choice was clear. (**) 

 

Examples:  using pi as initial seed calculate the first six RANs: 

You’d key:  PI, SEEDT, RNDM, RNDM, RNDM, RNDM, RNDM and RNDM  

To obtain:  



 

 

(*) The SandMath has since been patched with the same variation to remove the zeros. 

(**) We’ll make a more formal comparison between both cases later on, using the program “EVAL” 

Toulouse Math ROM Random Numbers 

The second set is a very interesting one, as it’ll be determined by the comparisons made for the 

“benchmarking” sections later on. It’s also interesting because the source was never released to the 

public - at least to my knowledge, pls. send feedback if you know better?  

Function Description Input Output 

STORAND Sets initial seed  Value in X Loaded in buffer 

RAND Randon Number Current RN in buffer Next RN in the sequence 

RANDXY RN between [a, b] Values in X,Y  Integer RN in [a,b] 

RCLRAND Gets current RN from buffer Current RN in buffer Current RN in X 

 

Here’s a short description for the functions in this group: 

STORAND is used to store the initial seed in the buffer. It uses Buffer #3, storing the seed value in 

digits <9:0> of the header register. This conflicts with the system standards used across the board and 

may cause havoc if you use other buffer utilities. BFCAT in particular will partially overwrite the RN 

value, as it uses the buffer header digits <2:0>; so better not to use it to be safe. 

RAND calculates the next RN in the sequence and stores its value as new seed 

RCLRAND recalls the current seed – not changing its value in the buffer. 

RANDXY provides a boundary for the resulting random number, which must be withing that interval.  

This implementation is also different (“C’est la France, vive la différence!” :-) in that it uses a digit level 

algorithm (see chapter on this later) instead of an LCG approach to generate the sequence of RNs. 

There’s a byte table in the code with values used to generate the RN digits according to the algorithm, 

still not sure how exactly but working on it as we speak. 

 

Same Example using pi as initial seed: 






 

And if now we want a RN between, say 12 and 17: 

12, ENTER^, 17, RANDXY =>  

 

This last function comes very handy for your Lottozahl needs (a.k.a “Lotería Primitiva”) – assuming 

your faith in science goes that far of course ;-) 

Voyager Random Numbers 

The RANROM module includes a version of the Voyager implementation of random numbers capability, 

taken from the VORANOGE-2 ROM,  prepared by mike-stgt  and published here:  

https://forum.hp41.org/viewtopic.php?f=13&t=428#p1279 

The use in the RANDOM module is covered under the Q Public license, see here for details. A copy of 

the license document is included as attachment to this manual. 

The functions included are shown below: 

Function Description Input Output 

SEED Sets initial seed  Value in X Loaded in buffer 

RN Randon Mumber Current RN in buffer Next RN in the sequence 

RCLSD Gets current RN from buffer None Current RN in X 

 

This set operates very much like the previous two cases for the SandMath and Toulouse Math ROM 

implementations. The values are stored in a buffer (with id# = 4) and like the Toulouse Math case – 

and contrary to the HP-41 default standards – the values are store in the buffer header itself, 

specifically in digits <9:0> (i.e. 10-digit mantissa format). This criterion was probably used by HP to 

save RAM memory, which was more at a premium in the 11C and 15C than in the HP-41C. 

In addition to the conflict mentioned in the previous case with BFCAT (overwriting of the current RN), 

here it gets more interesting because it will show this buffer as non-committed, i.e. unclaimed by any 

module and thus will be erased by the OS next time the calculator does a power off/on cycle. 

 

With that out of the way, the actual RNG scheme is given by the expression below: 

 

Which is another case of a Linear Congruent Generator (LCG) as defined here: 

https://en.wikipedia.org/wiki/Random_number_generation 

As far as the functions are concerned, we have a repeat (although chronologically speaking they came 

earlier) of the buffer-based design, with an initial seed function and a RAN instance for the elements of 

the sequence. RCLSD is a handy twist that can be used to know the current term in the sequence 

without generating the next one (for instance to make sure the seed is the desired one). 

 

Same example again, with pi as initial seed: 






 

 

https://forum.hp41.org/viewtopic.php?f=13&t=428%23p1279
https://opensource.org/licenses/QPL-1.0

Digit-Level Random Number Generator 

The functions below were written by Mark Power, an old hand in MCODE and active DataFile member, 

with many remarkable feats under his belt – such as the MCODE Debugger and PLAY ROMs.  

See http://www.hp41.org/LibView.cfm?Command=View&ItemID=1471  

and http://www.hp41.org/LibView.cfm?Command=View&ItemID=1472 

Function Description Input Output 

RAN00 RAN from seed in R00  Value in R00 Next RN in the sequence 

RAN20 RAN from seed in R20 Value in R20 Next RN in the sequence 

RANR _ _ RAN from sed in prompt reg Value in Rnn Next RN in the sequence 

 

These functions were published in DataFile V6N8 p9. They expect the seed (or previous RN) stored in 

the corresponding data register. The result is placed in X and stored in the same register upon 

completion.  

Examples. Storing pi in R00 we’ll key: 

PI, STO 00, RND00, RND00, RND00, RND00, RND00, RND00 

To get:  

  

RANR is a prompting function, accepting also INDirect registers – but not stack registers, sorry. 

     

In manual mode just enter the register number where you seed is stored. In a running program it will 

take it from the X-register instead, so it’ll be RANR IND X so to speak. 

 

Time-Based Seed Generator 

This function was written by Jean-Marc Baillard, see: http://hp41programs.yolasite.com/alea.php 

Use it to generate a time-based seed that can be used to generate a sequence of RN’s by any of the 

methods reviewed so far, (perfect complement for RAN00/RAN20 indeed) and a few more still to cover 

in the manual. 

Example: 

XEQ “1RAN”     =>  

 

Note: don’t use 1RAN to generate a sequence of RNs; doing so will generate skewed results due to the 

fairly similar “seeds” used – all taken from the internal Timer registers, which can be very similar in 

cases of TURBO running programs! 

 

http://www.hp41.org/LibView.cfm?Command=View&ItemID=1471%20
http://www.hp41.org/LibView.cfm?Command=View&ItemID=1472

MCODE listing for RAN00, RAN20, and RANR. 

Other Random Number Generators for the HP-41  

 This section is taken from Jean-Marc Baillard web pages, see: 

http://hp41programs.yolasite.com/alea.php 

Overview 

 Several pseudo random number generators are listed on this web page: 

"RNG1" "RNG2" "RNG3" work on every HP-41. 

"RNG4" & the M-Code routine 1RAN require a Time-Module. 

Finally, the last program is an attempt to play ( win? ) the lottery... 

 

Program #1  (35 bytes / SIZE 001) 

A well-known RNG is given by the formula:  Xn+1 = FRC ( 9821 Xn + 0.211327 ) which provides 1 

million random numbers. 

The following program gives 1,000,000,000  random numbers r  ( 0 <= r < 1 ). The formula  used is: 

Xn+1 = FRC ( 98 Xn + 0.236067977 )  

The coefficient 98 = 43,046,721 may be replaced by  a   where  a = 1 ( mod 20 ) 

and  the value in line 0.236067977  may be replaced by  b  where   b*109 is not divisible by 2 or 5. 

  

01  LBL "RNG1" 
02  9 
03  ENTER^ 
04  ENTER^ 
05  R^ 
06  *   
07  FRC             
08  * 
09  FRC 
10  * 
11  FRC 
12  *   
13  FRC             

14  * 
15  FRC 
16  * 
17  FRC 
18  *   
19  FRC             
20  * 
21  FRC 
22  5 
23  SQRT 
24  +   
25  FRC             
26  END 

 

       STACK    INPUTS     OUTPUTS 

           X           xn             xn+1 

 

Example: 

0.2   XEQ "RNG1"  yields    

R/S           ... etc ... 

  

http://hp41programs.yolasite.com/alea.php

Program #2    ( 26 bytes / SIZE 001 ) 

"RNG2" provides 9,999,999,996 random numbers with the formula: Xn+1 = ( 1059 Xn ) MOD p   

where p = 9,999,999,967 is the greatest prime < 1010 

Xn are integers between 0 and p (exclusive) which are then divided by p to be reduced to a number 

between 0 and 1. 

This routine works well because the MOD function gives exact results even when the operands are 

greater than 1010. 

Actually, the exponent E59 in line 3 may be replaced by any integer m provided m is relatively prime to  

p-1 = 2*3*11*457*331543, but I don't know what the best choice is. 

Unlike "RNG1" and other routines based upon the same type of formulae, the least significant digits 

don't go through any cycle of ten, one hundred and so on. 

Register R00 is used to store the different xn integers. 

  

01  LBL "RNG2" 
 02  RCL 00   STACK       INPUTS     OUTPUTS 

 03   E59       X               /           0 < r < 1 

 04  * 

 05  10 

 06  10^X 

 07  33 

 08  - 

 09  MOD 

 10  STO 00 

 11  LASTX 

 12  / 

 13  END 

 

Example: 

1  STO 00    XEQ "RNG2" gives      R00 = 3129146787 = mod (1059, p) 

                  R/S                           R00 = 6904570181  ... etc ... 

 

These ideas may be used to create your own RNG. 

Actually if p is a prime,  ( Z/pZ-{0} ; * ) is a group and if a is an integer,  the number of distinct 

elements in the subset   { 1 ; a ; a2 ; ....... ; ak ; .... } ( mod p )   divides  p-1 

If  p-1 is the smallest positive integer q such that aq = 1 ( mod p ) , then the sequence   

a ; a2 ; ....... ; ak ; .... ; ap-1 ( mod p )   is a permutation of  1 ; 2 ; ...... ; p-1 

In particular, if  p = 2p' + 1  where p' is also a prime, and if ap' is not equal to 1 ( mod p )  then  a  

satisfies the required property. 

For instance,  p = 7,841,296,787 = 2*3,920,648,393 + 1               

7,841,296,787 and 3,920,648,393 are primes and  -1024 = 4,851,307,369  ( mod p ) 

satisfies  (-1024)p' = -1  therefore the routine below gives  7,841,296,786  random integers 
{  E24,   *,    CHS,   7841296787,   MOD }                  
 

Program #3  ( 17 bytes / SIZE 001 ) 

The following algorithm is given by Clifford Pickover in "Keys to Infinity" ( John Wiley & Sons )  ISBN 0-

471-11857-5 

 01  LBL "RNG3" 
 02  LN         STACK  INPUTS      OUTPUTS 

 03   E2              X      xn           xn+1 

 04  * 

 05  1 

 06  MOD 

 07  END 

 

Example: 

    0.1   XEQ "RNG3" produces    

                   R/S                         ... etc ... 

 

Program #4  ( 25 bytes / SIZE 000 ) 

 01  LBL "RNG4" 
 02  DATE   STACK    INPUTS    OUTPUTS 

 03  TIME                  X             /        0 < r < 1 

 04  + 

 05   E49 

 06  * 

 07  PI 

 08  MOD 

 09  LN1+X 

 10  R-D 

 11  FRC 

 12  END 

No examples can be provided since the result depends on the instant you press R/S 

M-Code Routine 

This M-Code routine uses the TIME module - or an HP41-CX 

Monte Carlo Methods 

This chapter covers the application of Monte Carlo method to do diverse calculations, like integrals of 

functions of several variables, Mandelbrot set Area estimation, and approximations for numerical 

constants, such as Ln2, pi and e.  If there’s something they all have in common it’s their long execution 

time (ger ready to use V41 in TURBO mode) and the relative small accuracy of the results – unless 

many more iterations are done, which worsens the execution time, Be that as it may, this is an 

appropriate subject for the random topic so ready or not, here it comes. 

 

1. Monte Carlo Integration. 

The RANROM includes specific routines for the cases of functions of one, two, and three variables; as 

well as a general case for n-variables written by Greg McClure 

Function Description Input/Output Author 

“MCITG+ Driver for MCITG Prompts for params Ángel Martin 

“MCITG MC Integration – One var. FNAME in ALPHA Ángel Martin 

 “FX” Example function  n/a Á. Martin 

“MCITG3 MC Integration, up to 3 vars Prompts for params Ángel Martin 

 “FXY” Example with 2-vars n/a Á. Martin 

 “FXYZ” Example with 3-vars n/a Á. Martin 

“MCITGN MC Integration, n-vars See below. Greg McClure 

 

Let’s see a brief description for them, starting from the top.- 

MCITG and MCITG+ deal with integration of functions of a single variable (such as the example 

provided “FX”). It’s the simplest case but also the most likely to be used thus the dedicated routine for 

it. The driver function does all the parameter prompting for the user, and then calls the main 

subroutine – which can also be run separately provided that the user sets all those parameters 

manually prior to the call. 

The basic formula involves a repeat application of RNs {Xi}, at a very large scale: 

 

The program uses the SandMath-type p-RNG, i.e. functions SEEDT an RNDM. 

So let’s integrate f(x) = x^2 . e^x  between [0,1] using different number of iterations to see how that 

influences the result. 

XEQ “MCITG+” 

0, ENTER^, 1, R/S 

100, R/S    ; (ALPHA is turned ON) 

“FX”, R/S     ;  quite clearly we need more points… 

If flag 10 is set the program will display a countdown showing the current iteration, decreasing to 

“1”before presenting the result. 

Let’s repeat the integration using 1000 iterations as follows: 

1000  XEQ B   =>  

The correct result is shown below: 

  

 

 

MCITG3 can handle functions of up to three variables, thus the number of variables is also an input 

parameter to be entered – as well as the integration limits for each dimension and the number of 

iterations to run. The program will prompt for the input date so no need to set them up in advance. 

You can use MCITG3 for functions of a single variable, although MCITG will be easier and slightly 

faster. 

Let’s use MCITG3 to calculate an approximation of the provided functions FXY and FXYZ, between 

the intervals [0,1] in each dimension. 

f(x,y) = x^2 + y^2 

   

 

XEQ “MCITG3”  => 

2, R/S   =>  

0, ENTER^, 1, R/S => 

0, ENTER^, 1, R/S =>   ; (ALPHA is turned ON) 

“FXY”, R/S  =>  

1000, R/S  =>    , not quite good, try w/ more runs: 

R/S (or XEQ B) => 

10,000 R/S  => , getting better…

R/S   => 

100,000 R/S  => 

Very long execution times with slow “convergence” (for the lack of a better word), but 

surprising nonetheless… at least good to have as “the last resort” when everything else fails! 

Going now for the 3D example: 

f(x,y,z) = x^2 + y^2 + z^2 

 

 

XEQ “MCITG3”  => 

3, R/S   =>  

0, ENTER^, 1, R/S => 

0, ENTER^, 1, R/S => 

0, ENTER^, 1, R/S =>  

“FXYZ”, R/S  =>  

1000, R/S  =>  , not quite good, try w/ more runs: 

R/S   =>  

10000, R/S  =>   - it didn’t get better! 

This is an unexpected result and *may* be related to the lackluster quality of the p-RNG. Further 

testing should be done using other pairs of { SEED, RAN } functions instead to see if that gets things 

back to the “logical” path, i.e. “the more number of points, the better accuracy”. 

 

 Program listings. 

 01*LBL "MCITG+" 

 02*LBL A 

 03 "a^b=?" 
 04 PROMPT 
 05 "N=?" 
 06 PROMPT 
 07 "FNAME?" 
 08 AON 
 09 PROMPT 
 10 AOFF 

 11*LBL "MCITG" 

 12*LBL B 

 13 ASTO 00 
 14 STO 01 
 15 STO 05 
 16 RDN 
 17 X<>Y 

 18 STO 02 
 19 - 
 20 STO 03 
 21 0 
 22 STO 04 
 23 SEEDT 
 24*LBL 00 

 25 RNDM 
 26 RCL 03 
 27 ABS 
 28 * 
 29 RCL 02 
 30 + 
 31 XEQ IND 00 
 32 ST+ 04 
 33 FS? 10 
 34 VIEW 01 

 35 DSE 01 
 36 GTO 00 
 37 CLD 

 38*LBL C 

 39 RCL 04 
 40 RCL 05 
 41 / 
 42 RCL 03 
 43 * 
 44 RTN 
 45 GTO B 

 46*LBL "FX" 

 47 X^2 
 48 LASTX 
 49 E^X 
 50 * 
 51 END 

01*LBL "MCITG3" 

 02*LBL A 

 03  E 
 04 STO 02 
 05 3 
 06 "DIM=?(1,2,3)" 
 07 PROMPT 
 08 X>Y? 
 09 GTO A 
 10 ST+ X 
 11 4 
 12 + 
 13  E3 
 14 / 
 15 5 
 16 + 
 17 STO 00 
 18*LBL 00 

 19 0 
 20 ENTER^ 
 21  E 
 22 "a^b(" 
 23 RCL 00 
 24 4 
 25 - 
 26 2 
 27 / 
 28 INT 
 29  E 
 30 + 
 31 AINT 
 32 RDN 
 33 "`)=?" 
 34 PROMPT 
 35 X<>Y 

 36 ST- Y 
 37 X<>Y 
 38 ST* 02 
 39 ABS 
 40 STO IND 00 
 41 ISG 00 
 42 X<>Y 
 43 STO IND 00 
 44 ISG 00 
 45 GTO 00 
 46 "FNAME?" 
 47 AON 
 48 STOP 
 49 AOFF 
 50 ASTO 01 

 51*LBL B 

 52 "#POINTS=?" 
 53 PROMPT 
 54 STO \ 
 55 X<> 01 
 56 STO M 
 57 CLX 
 58 SEEDT 
 59 STO 03 
 60*LBL 03 

 61 RCL 00 
 62 FRC 
 63 5 
 64 + 
 65 STO 00 
 66*LBL 01 

 67 RNDM 
 68 RCL IND 00 
 69 * 
 70 ISG 00 

 71 RCL IND 00 
 72 + 
 73 ISG 00 
 74 GTO 01 
 75 XEQ IND M 
 76 ST+ 03 
 77 VIEW N 
 78 DSE N 
 79 GTO 03 
 80 RCL 03 
 81 RCL 02 
 82 * 
 83 RCL 01 
 84 / 
 85 RCL M 
 86 STO 01 
 87 RDN 
 88 "MCIT=" 
 89 ARCL X 
 90 PROMPT 
 91 GTO B 

 92*LBL "FXYZ" 

 93 XEQ 05 
 94 X<>Y 
 95 X^2 
 96 + 
 97 RTN 

 98*LBL "FXY" 

 99*LBL 05 

100 X^2 
101 X<>Y 
102 X^2 
103 + 
104 END 

 

Note. The Central Limit Theorem establishes that the error in the calculation is proportional to 1/sqr(N), 

with N being the number of points used. This explains the poor accuracy results for the reduced sets 

used in the previous example. In practice we should be using N>= 1,000,000 for a decent 

approximation. 

Finally, MCITGN is the general-case for functions of N-variables. Unfortunately there’s no driver 

section (ran out of room in the ROM!), therefore all the parameters must be manually entered (yes, 

this can be onerous…) in the expected registers prior to calling the routine, as follows: 

•  ORDER SHOULD BE IN R00 

•  NUMBER OF ITERATIONS (N) SHOULD BE IN R01 

•  POINTER TO RANDOM REGISTERS WILL BE SAVED IN R02 FOR USERS 

•  NAME OF USER FUNCTION SHOULD BE IN ALPHA, IT WILL BE SAVED IN R03 

•  INTEGRAL SUM WILL BE SAVED IN R04 (AS WILL BE FINAL RESULT) 

•  LIMITS IN R05-R2N+4 

•  RANDOM VALUES FOR USER FUNCTION IN R2N+5 TO R3N+4 
 

MCITGN was written by Greg McClure and posted on the MoHP Forum here: 
https://www.hpmuseum.org/forum/thread-6311.html?highlight=montecarlo 

Let’s see an example of a quintuple integral next. 

f(x,y,z,u,v) = sqrt (6 - x^2 - y^2 - z^2 - u^2 - v^2) 

 

; EXAMPLE OF QUINTUPLE INTEGRAL OF SQRT(6-X*X-Y*Y-Z*Z-U*U-V*V) 

; X FROM 0 TO 0.7 

; Y FROM 0 TO 0.8 

; Z FROM 0 TO 0.9 

; U FROM 0 TO 1.0 

; V FROM 0 TO 1.1 

 

; ALPHA = "5DINT" 

; REGISTER 00 = 5 

; REGISTER 01 = N (10, 100, AND 1000 USED FOR RUNS BELOW) 

; REGISTERS 05, 07, 09, 11, 13 = 0 

; REGISTER 06 = 0.7 

; REGISTER 08 = 0.8 

; REGISTER 10 = 0.9 

; REGISTER 12 = 1.0 

; REGISTER 14 = 1.1 

 

01  LBL "5DINT" 

02  6 

03  RCL 15 

04  X^2 

05  RCL 16 

06  X^2 

07  RCL 17 

08  X^2 

09  + 

10  + 

- 

11  RCL 18 

12  X^2 

13  RCL 19 

14  X^2 

15  + 

16  - 

17  SQRT 

18  END 

https://www.hpmuseum.org/forum/thread-6311.html?highlight=montecarlo

; 9 RUNS FOR N=10:   1.150 1.229 1.193 1.179 1.193 1.194 1.204 1.189 1.174 

; 6 RUNS FOR N=100:  1.192 1.190 1.179 1.193 1.187 1.192 

; 4 RUNS FOR N=1000: 1.186 1.192 1.192 1.190 

 

Commented Program Listing. 

01 LBL "MCINT" 

02 CF 21        ; AVIEW will show countdown. 
03 CLX 
04 STO 04     ; clear sum 
05 SEEDT      ; Seed - randomize 
06 ASTO 03   ; save user function name 
07 RCL 01      ; save count in O(7) 
08 STO O 
09 RCL 00      ; get dimension 
10 ST+ X        ; double for # of regs for limits 
11 E3 
12 / 
13 E 
14 +               ; convert to ISG value 
15 4.004       ; bump to point to first LL 
16 + 
17 STO M       ; save in M(5) 
18 RCL 00       ; create pointer to random regs. 
19 E3 
20 / 
21 RCL 00 
22 ST+ X 
23 + 
24 + 
25 STO N        ; save in N(6) 
26 STO 02       ; for user 
; PRODUCE RANDOM VALUES 
27 LBL 00 

28 RCL IND M     ; LL(N) in X 
29 ENTER^          ; LL(N) in X and Y 
30 ISG M 
31 RCL IND M     ; UL(N) IN X, LL(N) IN Y AND Z 
32 X<>Y 
33 -                ; UL(N) - LL(N) in X, LL(N) in Y 
34 RNDM      ; RAND 
35 * 
36 +         ; rand value between LL(N)  UL(N) in X 

37 STO IND N 
38 ISG M         ; continue 
39 STO X 
40 ISG N          ; bump random reg. pointer 
41 GTO 00       ; continue until done 
42 RCL 00 
43 ST- N         ; reset limits and values counters 
44 ST- M 
45 ST- M 
; CALL USER FUNCTION, SUM IN R04 
46 XEQ IND 03 
47 ST+ 04 
48 VIEW O 
49 DSE O 
50 GTO 00 
; FINAL RESULT. ; CALC MULTIPLIER 
51 1                           ; init multiplier 
52  LBL 01 

53  RCL IND M     ; LL(N) IN X 
54  ISG M 
55  RCL IND M   ; UL(N) in X, LL(N) in Y 
56  X<>Y 
57  -                     ; UL(N) - LL(N) in X 
58  *                    ; new multiplier 
59  ISG M 
60  GTO 01 
; MULTIPLY RESULT, DIVIDE BY NUMBER OF 
POINTS 
61  ST* 04 
62  RCL 02 
63  ST/ 04 
; RESTORE ALPHA NAME, DISPLAY RESULT 
64  CLA 
65  ARCL 03 
66  RCL 04 
67  CLD 
68  END 

2. Approximating Math Constants – Monte Carlo method 

This section uses a variation of the Monte Carlo strategy to evaluate both pi and e. It’s not, however, 

based in circle relationships derived from randomly throwing needles or shooting at targets, but on 

probability theory instead. It was explained by Valentín himself in his  HP Challenge VA511 - 2020-03-

14 - SRC 006 Pi Day 2020 Special.pdf 

Quoting directly from that article: 

“It's quite simple, actually. My recent program is this: 
 

1  DESTROY ALL @ RANDOMIZE 1 @ FOR K=1 TO 5 @ N=10^K @ S=0 

2  FOR I=1 TO N @ IF NOT MOD(IROUND(RND/RND),2) THEN S=S+1 
3  NEXT I @ P=S/N @ STD @ DISP N, @ FIX 3 @ DISP 5-P*4 @ NEXT K 

 

which is computing the probability that the closest integer to A/B is even, where A and B are 

uniformly distributed random numbers in [0,1), as produced by the RND keyword. Each time 

the rounded value is even (i.e., it's 0 modulo 2) the number of favorable outcomes (S) is 

incremented by one (see line 2). After N tries have been sampled, the probability P for the even 

case will be the number of favorables outcomes (S) divided by the number of tries (N), thus we 

have the estimated probability P = S/N. 

But I know from theory that in the limit, for N -> Infinity, the exact probability P = (5-Pi)/4, so 

isolating Pi we have Pi = 5-P*4, which is displayed by the program in line 3 above.” 
 

Note that he goes on to include yet another possible approach, which results in an even shorter BASIC 

program. Here’s the explanation: 

“Now, my earlier program, the one-liner, namely: 
 
10 INPUT K @ N=0 @ FOR I=1 TO K @ N=N-MOD(IROUND(RND/RND),2) @ NEXT I @ DISP 1-4*N/K 

 

is computing the probability that the closest integer to A/B is odd, where A and B are uniformly 

distributed random numbers in [0,1), as produced by the RND keyword. Each time the rounded 

value is odd (i.e., isn't 0 modulo 2) the number of favorable outcomes (N) is decremented by 

one, and after K tries have been sampled, the probability for the odd case will be the number of 

favorable outcomes (-N) divided by the number of tries (K), thus we have the estimated 

probability P = -N/K. 
 

As the probability of the rounded division being either even or odd is 1 (certainty), the probability for 

the odd case is 1 minus the probability for the even case, thus it's P = 1-(5-Pi)/4 = (Pi-1)/4, so isolating 

Pi we have Pi = 1+4*P = 1+4*(-N/K) = 1-4*N/K, which is then displayed by the one-line program.” 
      
I chose to use the first approach in this module, partially because it also requires the IROUND function, 

and I was intrigued by it. I ended up writing a short MCODE utility for that purpose, which facilitates 

the porting of the BASIC code to HP-41 FOCAL, shown in next page.  
 

 

https://albillo.hpcalc.org/challenges/HP%20Challenge%20VA511%20-%202020-03-14%20-%20SRC%20006%20Pi%20Day%202020%20Special.pdf
https://albillo.hpcalc.org/challenges/HP%20Challenge%20VA511%20-%202020-03-14%20-%20SRC%20006%20Pi%20Day%202020%20Special.pdf

With regard to the e calculation, the source has also been Valentín’s HP Challenge VA030 - Short Sweet 

Math Challenge 25 San Valentin Special - Weird Math.pdf. In that thread there’s one section (the first 

“concoction”) about calculating a “weird limit” that can be used for the calculation of e (making the 

sum--to-exceed s=1). 

“The limit average count for the sum of a series of [0,1) uniformly distributed random numbers to 

exceed 1 is exactly e = 2.71828182845904523536+, the base of the natural logarithms, which is pretty 

"weird" and can be considered an analog of Buffon's Needle experiment to estimate the value of Pi. 

Here we don't throw needles on a grid but merrily add up random numbers keeping count and we get e 

instead.” 

“This is the general formula to numerically compute the theoretically exact value  and my simple 1-

line, 53-byte HP-71B program to instantly compute them given the sum to exceed: “  

 

1 DESTROY ALL @ INPUT X @ S=0 @ FOR K=0 TO IP(X) @ S=S+(K-X)^K/FACT(K)*EXP(X-K) @    

NEXT K @ DISP S 

 

For the porting we’ll certainly need the new IROUND utility and obviously capable random number 

capabilities, which shouldn’t be much of a problem using the SandMath’s functions SEEDT and RNDM. 

E’ll use a time-generated initial seed (input zero for SEEDT), and RNDM will do the work using the well-
known RNG recurrence:  

 

r(k+1) = FRC [ r(k) * 9,821 + 0.211327 ] 
 

A few results are given in the table below: 

 
Iterations MCE MCPI 

10  

100  

1,000  

10,000  

100,000  

1,000,000  

 

 
As you can see from the table results above both routines require a very large number of iterations to 

get to a reasonably accurate result, which of course was expected as “it ‘comes with the territory” 
when resorting to this type of approaches. See below for the actual program code. 

 

 
 

 

https://albillo.hpcalc.org/challenges/HP%20Challenge%20VA030%20-%20Short%20Sweet%20Math%20Challenge%2025%20San%20Valentin%20Special%20-%20Weird%20Math.pdf
https://albillo.hpcalc.org/challenges/HP%20Challenge%20VA030%20-%20Short%20Sweet%20Math%20Challenge%2025%20San%20Valentin%20Special%20-%20Weird%20Math.pdf

Note:- The poor-man version of IROUND would consist of setting FIX 0 before the LBL 11 loop, and 

adding an INT instruction after the division of both random numbers (i.e. replacing IROUND with INT). 
That’s almost equivalent but doesn’t handle the EVEN condition for the result, i.e. IROUND(5.5)=5 

whereas INT(4.5) in FIX 0 is equal to 4 instead. Not a show-stopper though, considering how unlikely it 

is to find such an occurrence amongst the hundreds of random points used by the routine. 

One more for the road. 
 
And as the adagio goes “there’s never two without three”, so let’s add a third constant to this section – 

namely Ln 2 – another proud member of the irrational family. 

 
The following is taken from this Albert Chan’s post in the MoHP forum. 
 
He uses another condition from probability, not so well-known: 

 

LN(2) = 2 * probability of integer part of RND/RND is odd 
 
10 INPUT K @ N=0 @ FOR I=1 TO K @ N=N+MOD(IP(RND/RND),2) @ NEXT I @ DISP 2*N/K @ GOTO 10 
 
Moreover, we can improve LN(2) estimate by scaling RND/RND: 
 
> 10 DEF FNL(K) @ N=0 
> 20 FOR L1=1 TO K @ N=N+MOD(IP(10*RND/RND),2) @ NEXT L1 
> 30 FNL=N/(5*K)+1501/2520 @ END DEF 
 
See below the HP-41 version of the code. The driver program just adds the prompts for the initial seed 
and the number of points, then fall into the main subroutine – that expects N in X and the seed in Y 
registers. 
 

 
 
 

Examples:  

 
PI , 500 , XEQ “MCLN2” => 

 

  

https://www.hpmuseum.org/forum/thread-13796-post-129470.html#pid129470

3. Mandelbrot set Area estimation 
 

Saving the best for last, here is a brilliant example of RN’s utilization provided by Valentín Albillo’s 

excellent articles on the estimation of the Mandelbrot set area on the HP-42 and Free42 (see here: 

HP Article VA040a - Boldly Going - Mandelbrot Set Area (42S).pdf) 

Quoting sections or copying parts of that article is bound to do the reader and the article itself a huge 

disservice, so you’re encouraged to read the original – included in this manual in its entirety. Thanks to 

Valentín for graciously granting permission to do so. 

Porting it to the HP-41 platform was relatively straight-forward, once the function set was enhanced to 

deal with the required utilities. Obviously the HP-41 has its own limitations compared to the HP-42S 

and more so to Free42, however it does a good-enough job aided by the 41Z_Complex Number 

Module, needed for the complex math functions required by the program. 

Here’s the program listing on the HP-41 w/ the 41Z Module. 

 

 01*LBL "MBA" 

 02 2.5 
 03 STO 06 
 04 2 
 05 STO 07 
 06 1.2 
 07 STO 08 
 08 .25 
 09 STO 09 
 10 1 
 11 SEEDT 
 12 "POINTS=?" 
 13 PROMPT 
 14 STO 04 
 15 STO 00 
 16 256 
 17 "#ITERS=?" 
 18 PROMPT 
 19 STO 05 
 20 CLX 
 21 STO 02 
 22 "EVERY=?" 
 23 PROMPT 
 24 STO 03 
 25 CF 21 
 26 "WORKING..." 
 27 AVIEW 
 28 CF 00 

 29 X#0? 
 30 SF 00 
 31*LBL 00 

 32 RCL 05 
 33 STO 01 
 34 FS? 00 
 35 XEQ 03 
 36 RNDM 
 37 RCL 06 
 38 * 
 39 RCL 07 
 40 - 
 41 RNDM 
 42 RCL 08 
 43 * 
 44 X<>Y 
 45 ZRPL^ 
 46 ZSIGN 
 47 ZENTER^ 
 48 RCL 07 
 49 - 
 50 Z- 
 51 ZMOD 
 52 RCL 09 
 53 * 
 54 Z<>W 
 55 ZMOD 
 56 X<>Y 

 57 RDN 
 58 X<Y? 
 59 GTO 02 
 60 SIGN 
 61 ZRUP 
 62 RCL Z 
 63 - 
 64 ZMOD 
 65 RCL 09 
 66 X>Y? 
 67 GTO 02 
 68 ZRUP 
 69 ZRPL^ 
 70*LBL 01 

 71 Z^2 
 72 Z+ 
 73 ZMOD 
 74 RCL 07 
 75 X<=Y? 
 76 GTO 04 
 77 ZRDN 
 78 LASTZ 
 79 DSE 01 
 80 GTO 01 
 81*LBL 04 

 82 ISG 02 
 83*LBL 02 

 84 VIEW 00 

 85 DSE 00 
 86 GTO 00 
 87*LBL 03 

 88 RCL 00 
 89 RCL 03 
 90 MOD 
 92 X#0? 
 93 RTN 
 94 CLA 
 95 RCL 04 
 96 RCL 00 
 97 - 
 98 X=0? 
 99 RTN 
100 AINT 
101 "`->" 
102 RCL 02 
103 AINT 
104 PROMPT 
105 RCL Y 
106 / 
107 6 
108 * 
109 "AREA=" 
110 ARCL X 
111 AVIEW 
112 END 

 

https://albillo.hpcalc.org/articles/HP%20Article%20VA040a%20-%20Boldly%20Going%20-%20Mandelbrot%20Set%20Area%20(42S).pdf

Gaussian Distribution of Random Numbers 

Let’s move to a different chapter of the module, dealing with Gaussian- (or Normal-) distributed 

random numbers. As a way of introduction, Normal distribution of random numbers play an important 

role in numerous science fields and therefore must be covered. 

The module includes two techniques to obtain gaussian random numbers, both using output from the 

SandMath/PPC p-RNG described earlier in the manual.  The techniques are the Box-Muller and the “12R 

minus six” methods, described below. 

Function Description Input Output 

BXMR Gaussian RN (Box Muller) SEED/RN in buffer Normal RN in X 

RANG Gaussian RN (12R-6) SEED/RN  in buffer Normal RN in X 

 

See: https://en.wikipedia.org/wiki/Box%E2%80%93Muller_transform 

and: https://mathworld.wolfram.com/Box-MullerTransformation.html 

Note that both functions generate GRNs “Nr” with “Standard” Normal Distribution, i.e. with mean 0 and 

variance 1. This can be “moved” to GRNs “Xr” of any Normal Distribution with mean  and variance  

using the relationship:  Nr = (Xr - ) /  ;  and thus:    Xr =  + .Nr 

 

The Box-Muller method uses two uniformly distributed RN’s { x1, x2 } generated via SEEDT, RNDM to 

calculate two GRNs { z1, z2 } using the following transform: 

 

Whereas the “12R-6” method uses the sum of twelve random numbers generated via SEEDT, RNDM p-

RNG, and subtracts the value 6 from the result to obtain the GRN. Simpler formula but more 

demanding requiring 6x more RNs. 

z = (x1 + x2 + x3 + … + x12) - 6 

Both functions are implemented in MCODE, and the execution times are very similar, 

Example. Using pi as initial seed, generate 6 GRNs for each method: 

PI, XEQ “SEEDT” =>  

        BXMR              RANGxxxxx   x   

 
 
 
 
 

 

You should be getting the exact same results as the Time seed was not used. 

https://en.wikipedia.org/wiki/Box%E2%80%93Muller_transform
https://mathworld.wolfram.com/Box-MullerTransformation.html

Digging deeper: which p-RGN to use? 

The MCODE functions are “fixed” to the SandMath/PPC p-RNG but their FOCAL counterparts can use 

any of the three sets of p-RNGs, controlled by the user flags as follows: 

Function Description Input Output 

BX-MR Gaussian RN (Box Muller) SEED/RN in buffer Normal RN in X 

12R-6 Gaussian RN (12R-6) SEED/RN  in buffer Normal RN in X 

 

p-RNG used: 

Flag Set p-RNG Conditions 

UF 00 SandMath/PPC None 

UF 01 Toulouse Math UF 00 Clear 

UF 02 Voyager 11C/15C UF 00 and UF 01 clear 

 

Note that the interrogation follows the flag number order, thus if UF 00 is set that will prevail over the 

status of the other two because it’s the first one checked.  

Repeating last example for the other two p-RNG we obtain: (don’t forget to initialize the corresponding 

seed with PI, using the corresponding functions: STORAND and SEED  instead of SEEDT 

 

           For Voyager p-RNG – SF 02           For Toulouse Math p-RNG - SF 01      

          BX-MR               12R-6             BX-MR             12R-6  xxxxx      

 
 
 
 
 

 

 

And how to make heads or tails of all these GRNs? 

Armed with these two additional routines we can do comparative testing, pitting the methods against 

each other and choosing which p-RNG to use on every case. The routines “TRANG (Test RANG) and 

“BENCH are provided to that end, as described next. A third program “INDEX is also provided to 

evaluate the goodness of each combination of method and p-RNG – by establishing a comparison with 

the Normal Probability Function P(x). 

Normality Tests  Description Input  

“BENCH Driver for “TRANG” + “INDEX” Sample Size, p-RNG, Method  

“INDEX Calculate Normality Index Quantile data in {R00-R05}, R10  

“TRANG Test one Method/p-RNG combo Sample Size & User Flag 0-3  

 

Here’s how these work.-  

TRANG. Besides selecting the appropriate p-RNG user flag, “TRANG uses UF 03 to choose which 

method to employ: UF 03 Set = Bix-Muller, and UF 03 Clear = 12R-6 method.  

Once those four flags are properly set you would execute “TRANG to calculate the 6 quantiles reflecting 

the GRN probability distribution amongst them.  The width of the quantiles is determined by the sample 

size: w = N/6 

A final message in ALPHA like the one below shows the results for the first four quantiles – (not six, 

need to make it fit in ALPHA), the most important ones since usually the two remaining ones are 

always 100% (notice no decimal digits is provided for this cas to save some real state). 

For example, for a sample size of 50 GRNs and setting UF 02 (Voyager p-RNG) and also setting UF 03 

(Box-Muller method) we’ll type: 

XEQ “TRANG” =>   

50, R/S  => 

                                    

 

Is that good or bad? Well, here’s where function INDEX comes to the scene. It expects the quantile 

GRN probability data already stored (either by TRANG or manually by the user), and compares those 

percentages with the theoretical cumulative probability corresponding to a normal distribution, i.e. 

 

With x being the five abscissas of the five quantiles, and  the mean and standard deviation of the 

distribution. 

The module includes the Error Function erf(x) to calculate that integral by the following expression: 

 

The last step is a direct comparison between the theoretical and actual results, i.e. the “Normality 

Index”. For this example: 

XEQ “INDEX” =>    =>      

Calculated as: 

Index =  [sqrt (Q(n)^2 – P(n)^2],  n= 1,2…,6 

 A perfect set result would have index = 0, so the lower the better. 

All together now: Driver program 

For maximum convenience the driver program “BENCH” does the leg work with the User flag setting for 

you, navigating the different options with ALPHA prompts, see next. 

Let’s find the normality index of a set of 100 gaussian random numbers generated using the 12R-6 

method with the Toulouse Math p-RNG. We’d type: 

XEQ “BENCH” presents the first menu-driven screen with the name of the two methods: 

   choose [B] or [D] 

XEQ [D], which presents the second selection screen with the name of the three p-RNG choices: 

 choose [A], [C] or [E] 

XEQ [E] for the Toulouse Math, => 

100, R/S   => 

Set UF 10 to see a countdown with the current GRN being calculated. When completed we’ll see the 

“RUNNING. . .” message and then the ALPHA screen with the GRN percentages for each quantile: 

; 68.00; 95.00; 100; 100 ; 100; 100 

pressing R/S   =>   

 

Meaningful comparisons should be made between sets with equal sample sizes, or else the normality 

will be totally skewed of course.  

 

And the winner is… 

The table below summarizes the results like those obtained in the example above, (sample size = 100), 

repeated for all possible combinations. The initial seed is always 1. 

 

Method / p-RNG RNDM RN RAND 

Box-Muller   

12R-6   

 

The Toulouse Math ROM takes the gold, and that’s using either of the two methods. Second place is for 

the Voyager 11C/15C p-RNG, and the bronze medal is for the SandMath/PPC contender. 

Testing Random Number Generators 

 

There’s quite an abundance of relevant literature on this subject available on the internet. If you google 

the title of this chapter, the hit list is long, and the entries are quite informative, ranging from very 

pragmatical to heavy on the theoretical side. See for instance this one by Dan Biebighauser.  

For the RANROM I decided to take the historic perspective, including two old programs on the direct 

subject plus a more recent one about a related topic published in the MoHP forum. 

Program Description Author / Source  

“EVAL” Evaluation of p-RNGs L. H. Gilbert / UPL #10240  

“RNGTST” Randomicity Test Paper Charles T. Tart / Paper  

“CHI” Chi-square test   

“STRAT” Stratified Random Samplimg Rawi / MoHP Article  

 

1. Evaluation of p-RNGs   (by L.H. Gilbert, UPL# 10240) 

 

The original program has been modified slightly to allow for a more convenient handling of the p-RNG 

selection, which obviously cannot be entered in the main body of the program when this is in ROM. 

Besides I’ve replaced the data entry section with a new, menu-driven one. No other changes were 

made. Quoting from the original UPL document. - 

The program executes a user-provided random number generator, and checks the output in the 

following ways:  

- the mean and standard deviation of a sequence are computed 

- The correlation between Xi andX(i+1) is calculated 

- A running check is made to determine when the generator enters a closed cycle (if ever) 

- A histogram is plotted, and 

- A Chi-square test of uniformity is calculated. 

The program requires one memory module. A printer is recommended. 

Application, Equations, Variables. 

The random number generator produces a sequence of (uniform) numbers, X = { xi },  i = 1,…, N. 

- For each xi, x2i is computed and matched. If they’re equal, a closed cycle of numbers has been 

detected. 

- A histogram of 20 “bins” on the interval 0 – 1 is constructed, and a count made of where each 

xi falls. 

- The extended count per bin, N/20, is employed in a Chi-Square test: 

 ^2 =  [(fi – fe)^2 / fe    ; i= 1,2…,20 

 

with fi = expected count, and fe = N/20  

https://www.google.com/url?sa=t&rct=j&q=&esrc=s&source=web&cd=&ved=2ahUKEwjcmY-2ldn2AhWMKewKHZALAxMQFnoECDYQAQ&url=https%3A%2F%2Fwww-users.cse.umn.edu%2F~garrett%2Fstudents%2Freu%2FpRNGs.pdf&usg=AOvVaw0SGlZbT4zTsAPEpZAUlnw4

At the 5% level of significance the ^2 should be greater that 30 in order toconclude that the N random 

numbers are not uniformly distributed. 

 

Operating limits and warnings. 

The program is designed to evaluate a generator of uniform random numbers, on the interval [ 0, 1 [ 

Using all the program fcacilities, each number takes about 3.6 seconds to generate and evaluate; about 

3.0 seconds with the printer switched off. 

The RNG must be program under a general label in memory, it will be called by the main program. 

Example. 

A sequence of N=100 numbers was generated by the formula: Xi+1 = FRC [ 9821 * xi + 0.211327 ] 

The initial seed was 0.159753, and the formula has been programmed as: 

01 LBL “RNG0” 

02 9821 

03 * 

04 ,211327 

05 + 

06 FRC 

07 RTN 

Starting the program now: 

XEQ “EVAL” – sets SIZE 36 and resets all flags. 

The initial set of questions prepare the flags for the program execution automatically. You just need to 

enter the initial letter of the choice you want, and then press R/S. A shortcut is also enabled so if no 

letter is pressed the default option is the first choice.  

 

 R/S  - Clears/Sets UF 00

 R/S  - Clear /Sets UF 01

 R/S  - Clear/Sets UF 02  

 "H”, R/S - Clear/Sets UF 03

  R/S  - Clear Sets UF 04  

  “RNG0”, R/S 

   0.159753, R/S 

   100 , R/S 

The execution starts, showing a countdown as the sample is being processed. Finally, the HP-41 beeps 

and halts at end. Pressing R/S again starts the output of result – ending with the value of ^2 in the 

display. 

Here are the frequency counts on each bin for the first (left) and second (right) 100 digits: 

 

 

• If F00 is clear the execution halts after N numbers so that the printer can be switched off. If it is 

set no halt occurs, the printer is always on. 
 

• If F01 is clear each sequence of N numbers is treated separately,. If it’s set the data is 

accumulated for each successive sequence. 
 

• If F02 is clear the program, will check for duplicate numbers. If set there won’t be checked 0 

faster running time. 
 

• If F03 is clear the histogram is printed. If set, there’s no print of histogram. 
 

• If F04 is clear the program does a complete evaluation. If it’s set only a check for duplicate 

numbers is done. 

When a duplicate number is detected the program halts showing a  message that identifies the element 

in the sequence causing the event::  “”, 

Let’s now run EVAL for our three p-RNGs to – finally! – establish a comparison and therefore 

determine their relative ranking. For that we need to write three trivial FOCAL routines for EVAL to call, 

as follows: 

01  LBL “RPPC”  01  LBL “RTOU”  01  LBL “RVYG” 

02  RNDM  02  RAND  02  RN 

03  RTN   03  RTN   03  RTN 

Using a sequence length of N=1,000 and the same seed = 0.123456789 for all of them. 

The results are shown below: 

XROM "EVAL" 

  S RUN 

  A RUN 

  RUN 

  O RUN 

   RUN 

  “RPPC” RUN 

   0.123456789 RUN 

    1,000.000000 RUN 

N=1,000. 

SEED<N>= 1.234567890-01 

SEED<2N>=1.234567890E-1 

= 0.4954 s= 0.2791 

r=-0.0088 N= 1,000. 

E(f)= 50.0 

 44. 47. 50. 56. 54. 39. 56. 45. 49. 63. 69. 50. 45. 52. 51. 51. 44. 50. 42. 43. 

 

CHI-SQ<5%:30> = 19.80 

 

For the second group we use LBL B - there’s no need to repeat all data entry: 

XEQ C 

  “RTOU” RUN 

   0.123456789 RUN 

    1,000.000000 RUN 

 = 0.5152 s= 0.2884 

r=-0.0009 N= 1,000. 

E(f)= 50.0 

43. 53. 56. 45. 36. 49. 40. 51. 41. 59. 51. 44. 59. 56. 58. 52. 51. 56. 44. 56. 

 

CHI-SQ<5%:30> = 18.20 

And  going for the third group now: 

XEQ C 

RNG PRGM?  “RVYG” RUN 

SEED=?   0 .123456789 RUN 

N=?   1,000.00 RUN 

N=1,000. 

SEED<N>= 1.234567890-01 

SEED<2N>=1.234567890E-1 

 

 = 0.4918 s= 0.2837 

r=-0.0304 N= 1,000. 

E(f)= 50.0 

43. 46. 54. 57. 44. 50. 47. 56. 51. 56. 49. 66. 45. 44. 51. 41. 60. 51. 47. 42. 

CHI-SQ<5%:30> = 16.44 

 

Alright then, according to the results above the better p-RNG is the Voyager’ style. 

Uniformity Test SandMath/PPC Topulouse Math Voyager 11C/15C 

^2   

 

This is an interesting result, because despite being the better of the three in uniformity, if you recall it 

however did not hold the winner place for the normality of the gaussian random numbers generated 

from it; so not quite a slam dunk!  If anything, this demonstrates that there are different considerations 

to the quality of “randomness”. 

Going the extra mile, we can widen the comparisons by also testing some of the other p-RNGs available 

in the module, like Mark Power’s from DataFile and JM Baillard’s from his web site. Doing so we obtain 

the following table of results: 

 

Uniformity Test RAN00 RNG1 RNG2 

^2   

Program listing. 

01*LBL "EVAL" 

 02 SIZE? 
 03 37 
 04 X>Y? 
 05 PSIZE 
 06 CLX 
 07 X<>F 
 08 AON 
 09 "STOP/CONT?" 
 10 PROMPT 
 11 ATOX 
 12 83 
 13 X#Y? 
 14 SF 00 
 15 
"CLRG/ACCUM?" 
 16 PROMPT 
 17 ATOX 
 18 67 
 19 X#Y? 
 20 SF 01 
 21 
"XN=2N/OMIT?" 
 22 PROMPT 
 23 ATOX 
 24 88 
 25 X#Y? 
 26 SF 02 
 27 "HISTO/OMIT?" 
 28 PROMPT 
 29 ATOX 
 30 72 
 31 X#Y? 
 32 SF 03 
 33 "ALL/DUP?" 
 34 PROMPT 
 35 ATOX 
 36 65 
 37 X#Y? 
 38 SF 04 
 39 AOFF 

 40 REG 09 

 41*LBL C 

 42 CLRG 
 43 "RNG PRGM?" 

 44 AON 
 45 PROMPT 
 46 AOFF 
 47 ASTO 36 
 48 "SEED=?" 
 49 PROMPT 
 50 X=0? 
 51 1RAN 
 52 STO 00 
 53 STO 01 
 54 FS? 04 
 55 GTO 03 
 56 "N=?" 
 57 PROMPT 
 58 STO 02 
 59*LBL 01 

 60 RCL 00 
 61 RCL 01 
 62 RCL 02 
 63 RCL 36 
 64 FC? 01 
 65 CLRG 
 66 STO 36 
 67 RDN 
 68 FIX 0 
 69 "N=" 
 70 ARCL X 
 71 ACA 
 72 PRBUF 
 73 STO 02 
 74 STO 03 
 75 RDN 
 76 STO 01 
 77 RDN 
 78 STO 00 
 79 SCI 9 
 80 "SEED<N>=" 
 81 ACA 
 82 ACX 
 83 PRBUF 
 84 "SEED<2N>=" 
 85 ARCL 01 
 86 ACA 
 87 PRBUF 
 88 16 

 89 STO 15 
 90 CF 21 
 91 FIX 0 
 92*LBL 02 

 93 VIEW 03 
 94 RCL 00 
 95 RCL 00 
 96 XEQ IND 36 
 97 STO 00 
 98 s+ 
 99 CLX 
100 .05 
101 / 
102 INT 
103 RCL 15 
104 + 
105  E 
106 ST+ IND Y 
107 FC? 02 
108 XEQ 04 
109 DSE 03 
110 GTO 02 
111 BEEP 
112 FC? 00 
113 STOP 
114 FC? 55 
115 PRBUF 
116 SF 21 
117 XEQ 05 
118 XEQ 06 
119 FC? 00 
120 STOP 
121 ADV 
122 ADV 
123 GTO 01 
124*LBL 03 

125  E 
126 ST+ 03 
127 VIEW 03 
128 RCL 00 
129 XEQ IND 36 
130 STO 00 
131 XEQ 04 
132 GTO 03 
133*LBL 04 

134 RCL 01 
135 XEQ IND 36 
136 XEQ IND 36 
137 STO 01 
138 RCL 00 
139 X#Y? 
140 RTN 
141 TONE 9 
142 "CYCLE AT " 
143 ARCL 03 
144 PROMPT 
145 RTN 
146*LBL 05 
147 FIX 4 
148 2 
149 ACCHR 
150 61 
151 ACCHR 
152 MEAN 
153 ACX 
154 2 
155 SKPCHR 
156 115 
157 ACCHR 
158 61 
159 ACCHR 
160 SDEV 
161 ACX 
162 PRBUF 
163 114 
164 ACCHR 
165 61 
166 ACCHR 
167 SDEV 
168 * 
169 RCL 09 
170 RCL 11 
171 * 
172 RCL 14 
173 / 
174 RCL 13 
175 - 
176 CHS 
177 RCL 14 
178  E 

179 - 
180 / 
181 X<>Y 
182 / 
183 ACX 
184 " N= " 
185 FIX 0 
186 ARCL 14 
187 ACA 
188 PRBUF 
189 RTN 
190*LBL 06 

191 CLX 
192 STO 08 
193 RCL 15 
194 RCL 15 
195 19 
196 + 
197  E3 
198 / 
199 + 
200 STO 03 
201 69 
202 ACCHR 
203 40 
204 ACCHR 
205 102 
206 ACCHR 
207 41 
208 ACCHR 
209 61 
210 ACCHR 
211 RCL 14 

212 20 
213 / 
214 FIX 1 
215 ACX 
216 FIX 0 
217 RND 
218 STO 07 
219 STO 05 
220 PRBUF 
221 FS? 03 
222 GTO 08 
223 RCL 14 
224 2 
225 / 
226 SQRT 
227 RND 
228 ST+ 05 
229 - 
230 X<=0? 
231 0 
232 STO 04 
233 ADV 
234 ACX 
235 5 
236 SKPCHR 
237 "FREQ" 
238 ACA 
239 4 
240 SKPCHR 
241 RCL 05 
242 ACX 
243 ADV 
244 "------" 
245 ASTO L 

246 ARCL L 
247 ARCL L 
248 ACA 
249 ADV 
250 126 
251 RCL 05 
252 RCL 04 
253 - 
254 / 
255 RCL 07 
256 RCL 04 
257 - 
258 * 
259 INT 
260  E3 
261 / 
262 126 
263 + 
264 STO 06 
265*LBL 07 

266 RCL 03 
267 INT 
268 RCL 15 
269 - 
270 5 
271 * 
272 CLA 
273 ARCL X 
274 ACA 
275 RCL IND 03 
276 RCL 07 
277 - 
278 X^2 

279 ST+ 08 
280 RCL IND 03 
281 ACX 
282 RCL 04 
283 RCL 05 
284 RCL 06 
285 STKPLOT 
286 ISG 03 
287 GTO 07 
288*LBL 09 

289 ADV 
290 RCL 07 
291 ST/ 08 
292 FIX 2 
293 "CHI-SQ" 
294 "`<5%:30> =" 
295 ACA 
296 RCL 08 
297 ACX 
298 PRBUF 
299 ADV 
300 RTN 
301*LBL 08 

302 RCL IND 03 
303 ACX 
304 RCL 07 
305 - 
306 X^2 
307 ST+ 08 
308 ISG 03 
309 GTO 08 
310 GTO 09 
311 END 

 

Note that the initial seed is stored in R00 when first entered by the user, and that R00 is used to store 

all the random numbers as they’re being generated in the sequence. This works well for data-register 

based p-RNGs but for buffer-based p-RNGs we may need to initialize them manually entering the seed, 

using the corresponding SEED function for each case. This is only needed if we want the results to be 

comparable across different p-RNG, of course. 

2. Randomicity Testing.  (by Charles T. Tart) 
 

This program is designed for INTEGER random numbers and thus cannot be used for our cases- 

nevertheless it’s got intrinsic value that warrant including in the RANROM. Here’s the complete contents 

of the paper, available here.  

 

https://s3.amazonaws.com/cttart/articles/april2013articles/Randomicity+Test+Program+for+Pseudo-Random+Number+Generator+Routines+on+The+New+HP-41C.pdf

3. Stratified Random Sampling (by rawi, MoHP Articles) 

see: https://www.hpmuseum.org/forum/thread-15709.html  

The following is taken verbatim from the MoHP post. 

 

This is my first program for the HP 41 since decades. So it may be that there are numerous ways to 

improve the code.  I have tested it on a DM 41X and a HP 41CL. 

 

What it does: 

 

The program helps to analyze stratified random samples. 

You can either put in the weights (i.e. the shares of the strata in universe) and the standard 

deviations within the strata or you can put in a density function and the weights of the strata and 

the standard deviations within the strata are computed. 

 

For the allocation of the sample on the strata you can put in numbers or let the program compute 

sample sizes for proportional or optimal allocation. 

 

Standard deviation of total mean is computed. 

 

Needs command “FINTG” from the SandMath Module (or INTEG from the Advantage Pac). 

 

Use of registers: 

 

R00 - Final result 

R01 - Name of density function global label (Alpha) 

R02 - Number of strata (maximum: 6) 

R03 - Sample size 

R04 - Lower limit of first stratum 

 

Stratum  1  2  3  4  5  6 

Upper limit  R05  R06  R07  R08  R09  R10 

Weights  R11  R12 R13  R14  R15  R16 

Std. dev.  R17  R18  R19  R20  R21  R22 

 

RegisterS R23-R30 are used for computations. 

 

The function FINTG from SandMath module needs another 32 unused program registers. 

Usage instructions: See program listing. 

 

 

 

https://www.hpmuseum.org/forum/thread-15709.html

Example: 

You want to analyze a stratified sample with 4 strata and optimum allocation of sample. The variable 

has a standard normal distribution. Strata limits are -1, 0, 1. Total sample size is n=1000. 

For convenience you take as lower limit of standard normal distribution -7.5 and as upper limit 7.5.  

First type in routine for normal distribution under a global label: 

01 LBL “NV” 

02 X^2 

03 2 

04 / 

05 CHS 

06 E^X 

07 2 

08 PI  

09 * 

10 SQRT 

11 / 

12 RTN

 

XEQ “STRAT” 

 

->     1000 ,  R/S (sample size input) 

->     4 ,  R/S (input of number of strata) 

->     1 , R/S (we use the densitiy function, so we type 1) 

->      “NV” , R/S (input of label of program with density function) 

->       7.5 CHS ,  R/S (lower limit of first stratum is -7.5) 

->       1 CHS ,  R/S  (upper limit of first stratum is -1) 

->       0 ,  R/S 

->       1 ,  R/S 

->       7.5 ,  R/S 

 

After about 15 minutes (HP 41) / 20 seconds (DM 41X with USB cable): 

->       2 ,  R/S (we want optimum allocation, so we type 2) 

->        (sample size in stratum 1)  

R/S ->     (sample size in stratum 2) 

R/S ->   

R/S ->   

R/S ->          (std deviation of stratified sample mean) 

 

Weights and standard deviations can be seen in registers, e.g. weight in stratum 1:  

RCL 11 ->  



Program listing. 

 

01    LBL STRAT 

02    ”SMPL SZ?” 
03    PROMPT            Input of sample size 
04    STO 03 
05    “N STRATA?” 
06    PROMPT            Input of number of strata (maximum: 6) 
07    1 E3 
08    / 
09    1 
10    + 
11    STO 02 
12    0 
13    STO 00 
14    STO 27 
15    “0=I 1=DF?”            Input whether weights and std. dev. of strata are individually typed 
16    PROMPT            (0) or whether they are computed by given density function (1) 
17    4 
18    + 
19    XEQ IND X 
20    “0=I 1=P 2=O?”        Input whether allocation of sample on strata should be individually 
21    PROMPT            given (0), proportional to weights (1) or optimal (2) 
22    STO 30 
23    FIX 0 
24    10 
25    STO 28 
26    16 
27    STO 29 
28    LBL 08            Determination of sample size in strata 

29    1 
30    ST+ 28 
31    ST+ 29 
32    RCL IND 28 
33    RCL IND 29 
34    XEQ IND 30 
35    RCL IND 28 
36    RCL IND 29 
37    * 
38    x^2 
39    x<>y 
40    / 
41    ST+ 00 
42    ISG 02 
43    GTO 08 

44    FIX 4 
45    RCL 00 
46    SQRT 
47    STO 00              Standard dev. of mean of total sample is shown 
48    CF 01 
49    RTN 
50    LBL 04              Manual Input of weights and std. dev. in strata 

51    10 
52    STO 28 
53    16  
54    STO 29 
55    FIX 0 
56    LBL 06 

57    1  
58    ST+ 28 
59    ST+ 29 
60    “W” 
61    ARCL 02 
62    “|-?” 
63    PROMPT 
64    STO IND 28 
65    “S” 
66    ARCL 02 
67    “|-? 
68    PROMPT 
69    STO IND 29 
70    * 
71    ST+ 27 
72    ISG 02 
73    GTO 06 
74    XEQ 09 
75    RTN 
76    LBL 05               Computation of weights and std. dev. in strata with given function 

77    AON                
78    FIX 0 
79    “NM DF?”          Asks for name of global program label with density function 
80    PROMPT 
81    ASTO 01 
82    AOFF 
83    4 
84    STO 28 
85    “LL S1?”             Asks for lower limit of stratum 1 
86    PROMPT 
87    STO 04 
88    STO 23 
89    LBL 07 

90    1 

91    ST+ 28 
92    “UL S” 
93    ARCL 02 
94    “|-?” 
95    PROMPT            Asks for upper limit of stratum l 
96    STO IND 28 
97    ISG 02 
98    GTO 07 
99    FIX 5                Format defines accuracy of integration 
100    4 
101    STO 28 
102    10 
103    STO 29 
104    XEQ 09 
105    LBL 10            Computation of parameters of strata 

106    1 
107    ST+ 28 
108    ST+ 29 
109    CLA 
110    CF 01 
111    ARCL 01 
112    RCL 23 
113    RCL IND 28 
114    STO 24 
115    INTEG             Computation of weight of stratum l = integral(f(x)) 
116    STO IND 29 
117    STO 25 
118    “STRAX” 
119    RCL 23 
120    RCL 24 
121    INTEG             Computation of integral(x*f(x)) 
122    RCL 25 
123    / 
124    STO 26            Mean in stratum 
125    SF 01 
126    RCL 23 
127    RCL 24 
128    STO 23 
129    INTEG             Computation of integral(x²*f(x)) 
130    RCL 25 
131    / 
132    RCL 26 
133    X^2 
134    - 
135    SQRT              std. dev. in stratum 
136    RCL 29 
137    6 



 

  

© ÁNGEL M. MARTIN – MARCH 2022 PAGE 57 OF 58 

  

     HP- RANDOM ROM  QRG 

138    + 
139    x<>y 
140    STO IND Y 
141    RCL 25 
142    *                   weight in stratum * std. dev. in stratum 
143    ST+ 27 
144    ISG 02 
145    GTO 10 
146    XEQ 09 
147    RTN 
148    LBL 09           Refreshing loop register 02 

149    RCL 02 
150    FRC 
151    1 
152    + 
153    STO 02 
154    RTN 
155    LBL 00           Manual input of sample size in stratum 

156    “N” 
157    ARCL 02 
158    “|-?” 
159    PROMPT 
160    RTN 
161    LBL 01          Computation of proportional sample size in stratum 

162    X<>Y             (share of stratum in sample and in universe are equal) 
163    RCL 03 
164    * 
165    GTO 03 
166    LBL 02     Computation of optimum sample size in stratum  

167    *              (proportional to weight*std. dev in stratum) 
168    RCL 27 
169    / 
170    RCL 03 
171    *  
172    LBL 03   Output of sample size in stratum 

173    AON 
174    “N” 
175    ARCL 02 
176    “|-=” 
177    ARCL X 
178    STOP 
179    AOFF 
180     END 

01    LBL “STRAX”      Subroutine for computation of integral x*f(x) and x²* f(x) 

02    STO 30                depending of status of flag 01 
03    XEQ IND 01 
04    RCL 30 
05    FS? 01 
06    x^2 
07    * 
08    END 

 

 

 

Appendix – Original DataFile Article by Mark Power 

Appendix.- Q Public Licence. 

 


